


CNAM - NSY116 – jan2015

ED 11. Captation

P. Cubaud

cubaud @ cnam.fr

Mon jeu incroyable



```
score = 1000;  
while (curseur_pas_sur_la_cible()) {  
    score--;  
    if (score<0) noLoop();  
}
```

```
int score=300;  
int RAYON = 100;  
int x,y;
```

```
void setup(){  
  size(400,400);  
  noCursor();  
  x = width/2;  
  y = height/2-100;  
}
```

```
void draw(){  
  // objet guide  
  background(200);  
  noFill();stroke(0);strokeWeight(10);  
  ellipse(width/2,height/2,2*RAYON,2*RAYON);  
  // le pointeur  
  x = mouseX;  
  y = mouseY;  
  fill(250,0,0);noStroke();  
  ellipse(x+random(-10,10),y+random(-10,10),30,30);  
  // le score  
  float d = sqrt((x-width/2)*(x-width/2)+(y-height/2)*(y-height/2));  
  if ((d < RAYON-5)|| (d > RAYON+5)) score--;  
  fill(0);textSize(40);textAlign(CENTER);text(str(score),width/2,50);  
  // arret du jeu  
  if (score==0) noLoop();  
}
```

pour
simuler
le capteur

Exercice : reprendre le code avec la captation video et localisation du pixel le plus rouge

```
monjeuVIDEO
int score=30000;
int RAYON = 100;
int x,y;
import processing.video.*;
Capture video;
float txe = 0;
float tye = 0;
float alpha = 0.10;


void setup(){
    size(400,400);
    noCursor();
    video = new Capture(this, width, height, 30);
    //video.start(); // pour Processing 2.0 et +
    x = width/2;
    y = height/2-100;
}
```

```
void draw(){
  background(200);
  //image(video, 0, 0, width, height);
  // objet guide
  noFill();stroke(0);strokeWeight(10);
  ellipse(width/2,height/2,2*RAYON,2*RAYON);
  // le pointeur
  curseurVideo();
  x = width-(int)txe;
  y = (int)tye;
  fill(250,0,0);noStroke();
  ellipse(x,y,30,30);
  // le score
  float d = sqrt((x-width/2)*(x-width/2)+(y-height/2)*(y-height/2));
  if ((d < RAYON-5)||(d > RAYON+5)) score--;
  fill(0);textSize(40);textAlign(CENTER);text(str(score),width/2,50);
  // arret du jeu
  if (score==0) noLoop();
}
```

(solution)

```
void curseurVideo(){
    float hmax = 0;
    int tx = 0;
    int ty = 0;
    video.read();
    video.loadPixels();
    int index = 0;
    for (int y = 0; y < video.height; y++) {
        for (int x = 0; x < video.width; x++) {
            float h = hue(video.pixels[index]);
            // la teinte max donne du rouge
            if (h > hmax){
                hmax = h; tx = x; ty = y;
            }
            index++;
        }
    }
    txe = (1-alpha)*txe + alpha*tx;
    tye = (1-alpha)*tye + alpha*ty;
}
```

VERSION 3 : AVEC OPENCV



C'est trop fatigant avec sa vraie tête



VERSION 4 : avec la leap motion

- récupérer les coordonnées de l'index (ou du doigt le plus proche de l'écran)

- voir le tuto

<http://tutoprocessing.com/avance/leap-motion-processing/>

- au lieu d'afficher tous les doigts comme dans ce tuto, rechercher celui de coordonnée z (?) le plus petit (grand ?)